Program Run Order:

sitc_deflatedpriceindex.sas


inputs:
 ~/rawdata/SITC_2digit_ForIndex/x_priceindex_96to99.txt



 ~/rawdata/SITC_2digit_ForIndex/SITC_2DIGIT_ANNUAL_95.TXT


output:
 ~/rawdata/xprice.ssd01

review_tables.sas ( Tables 1-6


inputs:
 ~/rawdata/asm_shipments.dat



~/rawdata/Employment/pop97.dat



~/rawdata/SIC_2digit/st96om.dat



~/rawdata/SIC_2digit/st97om.dat



~/rawdata/SIC_2digit/st98om.dat



 ~/rawdata/xprice.ssd01

review_regress.sas ( Regression Analysis


inputs:
 ~/rawdata/oilshock.dat



~/rawdata/Employment/manuemp_bystate_nsa.dat



~/rawdata/asm_shipments.dat

~/rawdata/SIC_2digit/st96om.dat



~/rawdata/SIC_2digit/st97om.dat



~/rawdata/SIC_2digit/st98om.dat



 ~/rawdata/xprice.ssd01

Create SIC Export Price Indices from Raw SITC Export Price Indices

--Change %let statement to reflect the last month of available data.

--Change the input statement by removing the $ from behind the variables that are newly available since the last run.

Output library:  xprice.ssd01

Program file:

sitc_deflatedpriceindex.sas

filename in 'd:\sas\state_exports\rawdata\SITC_2digit_ForIndex';

filename in2 'd:\sas\state_exports\rawdata';

libname out 'd:\sas\state_exports\rawdata';

*****************************************************************************;

**This is ~/state_exorts/sitc_deflatedpriceindex.sas                       **;

**The input files are ~/sic/rawdata/SITC_2digit_ForIndex/                  **;

**  x_priceindex_96to99.txt and SITC_2DIGIT_ANNUAL_95.TXT.                 **;

**This program creates SIC-based export price indices from raw SITC-based  **;

**  indices.  These price indices are used to deflate the raw export data  **;

**  to create the real percent changes reported in the tables of the       **;

**  Pollard/Coughlin Review article.                                       **;

**The 1995 nominal export levels are used to create weighted price indices **;

**  when you need to use a group of SITC indices to create the SIC index.  **;

**When creating a new year to date calculation, all you need to do is      **;

**  change the %let ytd= statement below to equal the number for the last  **;

**  month of available export level data.                                  **;

**This program was written by Heidi Beyer x8805                            **;

*****************************************************************************;

options pagesize=100 linesize=150 nocenter nodate;

*************************************;

**ENTER LAST MONTH FOR YEAR TO DATE**;

*************************************;

%let ytd = 12;

********************************************;

**PULL IN THE EXPORT PRICE LEVELS FROM BLS**;

********************************************;

data x_price(drop=nmq1_96 nmq2_96 nmq3_96 nmq4_96

                  nmq1_97 nmq2_97 nmq3_97 nmq4_97

                  nmq1_98 nmq2_98 nmq3_98 nmq4_98

                  nmq1_99 nmq2_99 nmq3_99 nmq4_99

                  _96m1-_96m12

                  _97m1-_97m12

                  _98m1-_98m12

                  ann96 ann97 ann98 ann99);

  infile in(x_priceindex_96to99) linesize=1000 firstobs=305 dlm=","  pad;

  input seriesid $ _96m1-_96m12 ann96 $13.

                   _97m1-_97m12 ann97 $13.

                   _98m1-_98m12 ann98 $13.

                   _99m1   _99m2  _99m3  _99m4  _99m5  $ _99m6  $

                   _99m7 $ _99m8 $ _99m9 $ _99m10 $ _99m11 $ _99m12 $ ann99 $ @;

*******************************************************;

**CHANGE 1999 MISSING VALUES FROM ": No data" TO "." **;

*******************************************************;

    %macro strange;

      %do i = 1 %to 12;

         if substr(_99m&i,7,2)="No" then _99m&i=".";

      %end;

    %mend;

    %strange;

  length sitccode 3.;

**********************************************************************;

**MAKE 2 DIGIT SITC CODES FOR THE 1-DIGIT MAJOR INDUSTRY GROUPINGS  **;

**********************************************************************;

    if compress(seriesid)="EIUIE1" then sitccode=1*10;

     else if compress(seriesid)="EIUIE2" then sitccode=2*10;

     else if compress(seriesid)="EIUIE3" then sitccode=3*10;

     else if compress(seriesid)="EIUIE4" then sitccode=4*10;

     else if compress(seriesid)="EIUIE5" then sitccode=5*10;

     else if compress(seriesid)="EIUIE6" then sitccode=6*10;

     else if compress(seriesid)="EIUIE7" then sitccode=7*10;

     else if compress(seriesid)="EIUIE8" then sitccode=8*10;

     else if compress(seriesid)="EIUIE9" then sitccode=9*10;

*******************************************************;

**MAKE SITC CODES FOR THE "OTHER" INDUSTRY GROUPINGS **;

*******************************************************;

     else if compress(seriesid)="EIUIE2R" then sitccode=222;

     else if compress(seriesid)="EIUIE8R" then sitccode=888;

****************************************************************;

**MAKE SITC CODE OUT OF THE LAST 2 DIGITS OF THE BLS MNEMONIC **;

****************************************************************;

     else sitccode = substr(seriesid,6,3);

**************************************************************;

**CREATE ANNUAL AND QUARTERLY AVERAGES FROM RAW MONTHLY DATA**;

**************************************************************;

    %macro chngfreq;

        %do i = 96 %to 99;

          **Create annual averages of available data;

            y&i=mean(of _&i.m1-_&i.m&ytd);

 %end;

    %mend;

    %chngfreq;

run;

***************************************************;

**PRINT AVERAGE PRICE INDICES FOR VARIOUS PERIODS**;

***************************************************;

proc print;

  var sitccode y96 y97 y98 y99;

  title "SITC Export Price Indices";

  title2 "(YTD Through Month=&ytd)";

run;

********************************************;

**PULL IN THE 1995 EXPORT LEVELS FROM ITC **;

**Use 1995 becasue the export prices have **;

**  a 1995 base period.                   **;

********************************************;

data x_levels(drop=sitccode);

  infile in(SITC_2DIGIT_ANNUAL_95) firstobs=3 dlm="," pad;

  input sitccode x95level;

  *Create groups for changeing SITC into SIC where there is no direct match between them;

    if sitccode in(0,2,4,5,21,22) then group1=x95level;

    if sitccode in(1,2,3,4,5,6,7,8,9,11,29,41,42,43) then group2=x95level;

    if sitccode in(23,62) then group3=x95level;

    if sitccode in(24,63,81) then group4=x95level;

    if sitccode in(25,64) then group5=x95level;

    if sitccode in(26,65) then group6=x95level;

    if sitccode in(51,52,53,54,55,56,57,58,59) then group7=x95level;

    if sitccode in(61,83,85) then group8=x95level;

    if sitccode in(67,68) then group9=x95level;

    if sitccode in(71,72,73,74,75) then group10=x95level;

    if sitccode in(76,77) then group11=x95level;

    if sitccode in(78,79) then group12=x95level;

    if sitccode in(87,88) then group13=x95level;

  *Assign SITC codes to use for industries where there is no 2-digit SITC export price index;

    if sitccode in(0,2,6,7) then sitc_new=0;

       else if sitccode in(11) then sitc_new=10;

       else if sitccode in(23,29) then sitc_new=222;

       else if sitccode in(34) then sitc_new=30;

       else if sitccode in(41,42,43) then sitc_new=40;

       else if sitccode in(61) then sitc_new=60;

       else if sitccode in(79) then sitc_new=70;

       else if sitccode in(81,83,85) then sitc_new=888;

      else sitc_new=sitccode;

    sitc_old=sitccode;

run;

********************************************;

**CREATE EXPORT LEVELS FOR GROUPS WITH NO **;

**  SITC=SIC DIRECT MATCH                 **;

********************************************;

proc means data=x_levels sum;

  var group1-group13;

  output out=grplevel sum=;

  title "1995 Export Levels by Groups of SITCs";

run;

********************************************;

**COMBINE GROUP LEVELS AND INDIVID LEVELS **;

********************************************;

data x_levels;

  if _n_=1 then set grplevel;

  set x_levels(keep=sitc_old sitc_new x95level);

run;

******************************;

**PRINT 1995 EXPORT LEVELS  **;

******************************;

proc print;

  var sitc_old sitc_new x95level;

  title "Export Levels by SITC Codes";

  title2 "Annual 1995";

run;

********************************************;

**CREATE SHARE INDEX FOR EACH SITC IN GROUP*;

**SHARE = PRICEINDEX * 1995LEVEL/GROUPLEVEL*;

********************************************;

proc sort data=x_price; by sitccode; run;

proc sort data=x_levels; by sitc_new; run;

data start;

  merge x_price x_levels(rename=(sitc_new=sitccode));

  by sitccode;

  %macro defgrp(sitcno=,gno=,freq=);

    %do i = 1 %to 54  ;

    %let g=%scan(&gno,&i);

    %let s=%scan(&sitcno,&i);

      %do j=1 %to 4;

      %let f=%scan(&freq,&j);

        if sitc_old = &s then g&g&f=&f*x95level/group&g;

      %end;

    %end;

  %mend;

  %defgrp(sitcno=0 2 4 5 21 22 1 2 3 4 5 6 7 8 9 11 29 41 42 43 23 62 24 63 81 25 64 26 65 51 52 53

                 54 55 56 57 58 59 61 83 85 67 68 71 72 73 74 75 76 77 78 79 87 88,

          gno=1 1 1 1 1 1 2 2 2 2 2 2 2 2 2 2 2 2 2 2 3 3 4 4 4 5 5 6 6 7 7 7 7 7 7 7 7 7 8 8 8 9 9

              10 10 10 10 10 11 11 12 12 13 13,

          freq=y96 y97 y98 y99);

run;

********************************************;

**SUM GROUP SHARES TO CREATE GROUP INDEX  **;

********************************************;

%macro grpindex;

  %do y = 96 %to 99;

    proc means data=start sum n;

      var g1y&y g2y&y g3y&y g4y&y g5y&y g6y&y g7y&y g8y&y g9y&y g10y&y g11y&y g12y&y g13y&y;

      output out=defy&y sum=;

      title "Export Deflated Export Price by Groups of SITCs 19&y";

    run;

%end;

%mend;

%grpindex;

********************************************;

**COMBINE GROUP DATA FROM ALL PERIODS     **;

********************************************;

data groups;

  merge defy96 defy97 defy98 defy99;

run;

***********************************************************;

**MATCH SIC TO SITC INDEXES WHERE THERE IS A DIRECT MATCH**;

***********************************************************;

data sic;

  input siccode sitccode;

  cards;

1      .

2      .

8      3

9      3

10     28

12     32

13     30

14     27

20     .

21     12

22     .

23     84

24     .

25     82

26     .

27     89

28     .

29     33

30     .

31     .

32     66

33     .

34     69

35     .

36     .

37     .

38     .

39     89

0102   .

0709   3

1112   32

;

proc sort data=sic; by sitccode; run;

*******************************;

**MATCH SITC DIRECTLY TO SIC **;

**Keeping only anns for now **;

*******************************;

data sitc_sic(keep=siccode sitccode y96 y97 y98 y99);

  if _n_=1 then set groups;

  merge x_price (keep=sitccode y96 y97 y98 y99)

sic;

  by sitccode;

*******************************;

**MATCH SITC GROUPS TO SIC   **;

*******************************;

  %macro tosic(vari=);

    %do i = 1 %to 4;

    %let v = %scan(&vari,&i);

      if sitccode<>"." and siccode<>"." then &v=&v;

        else if siccode in(1,2,102) then &v=g1&v;

        else if siccode in(20) then &v=g2&v;

        else if siccode in(30) then &v=g3&v;

        else if siccode in(24) then &v=g4&v;

        else if siccode in(26) then &v=g5&v;

        else if siccode in(22) then &v=g6&v;

        else if siccode in(28) then &v=g7&v;

        else if siccode in(31) then &v=g8&v;

        else if siccode in(33) then &v=g9&v;

        else if siccode in(35) then &v=g10&v;

        else if siccode in(36) then &v=g11&v;

        else if siccode in(37) then &v=g12&v;

        else if siccode in(38) then &v=g13&v;

    %end;

  %mend;

  %tosic(vari=y96 y97 y98 y99);

  if siccode="." or siccode=0 then delete;

run;

********************************************************;

**CREATE DATASET TO DEFLATE THE EXPORT LEVELS FROM ITC**;

********************************************************;

proc sort data=sitc_sic out=out.xprice; by siccode; run;

proc print;

  var siccode sitccode y96 y97 y98 y99;

  title "SIC Export Price Indices";

  title2 "(YTD Through Month=&ytd)";

run;
These are the details for creating the SIC-based export price indices.

SITC Groupings For Converting Price Index to SIC Based

Group1:
0,2,4,5,21,22

Group2:
1,2,3,4,5,6,7,8,9,11,29,41,42,43

Group3:
23,62

Group4:
24,63,81

Group5:
25,64
Group6:
26,65

Group7:
51,52,53,54,55,56,57,58,59

Group8:
61,83,85

Group9:
67,68


Group10: 71,72,73,74,75

Group11: 76,77

Group12: 78,79

Group13: 87,88



To calculate group index, sum 1995 exports for each group.  Then take each individual SITC code within in the group and perform the following calculation:  SITC# Price Index  *  SITC# 1995 Exports  /  Group# 1995 Exports.  Finally sum the previous calculation by group.

For Missing SITC Price Indices, here are the SITC substitutes:

0,2,6,7
 
0

11

10

23,29

2R

34                       30

41,42,43

40

61 60

79 70

81,83,85

8R

SITC Export Price Indices to Use For Each SIC Code

SIC

1

2

8

9

10

12

13

14

20

21

22

23

24

25

26
SITC Codes Used

0,2,4,5,21,22(G1)

0,2,4,5,21,22(G1)

3

3

28

32

34

27

1,2,3,4,5,6,7,8,9,11,29,41,42,43(G2)

12

26,65(G6)

84

24,63,81(G4)

82

25,64(G5)

SIC

27

28

29

30

31

32

33

34

35

36

37

38

39

01&02

07-09

11-12
SITC Codes Used

89

51,52,53,54,55,56,57,58,59(G7)

33

23,62(G3)

61,83,85(G8)

66

67,68(G9)

69

71,72,73,74,75(G10)

76,77(G11)

78,79(G12)

87,88(G13)

89

0,2,4,5,21,22(G1)

3

32



Create the Tables

Program File:


review_tables.sas

*****************************************************************************;

**This is ~/state_exports/Review_tables.sas                                **;

**The input files are                                                      **;

**      ~/state_exports/rawdata/SIC_2digit/ST97OM.DAT and ST98OM.DAT       **;

**      ~/state_exports/rawdata/Employment/pop97.dat                       **;

**      ~/state_exports/rawdata/asm_shipments.dat                          **;

**      ~/state_exports/rawdata/SIC_2digit/ST96OM.DAT                      **;

**The data library used is ~/sic/rawdata/xprice.ssd01.                     **;

**This program creates Tables 1-6 of the Pollard/Coughlin Review article.  **;

**When new data becomes available change %let goyr= to account for a diff. **;

**  base year for real data or shipments data and %let yr?= to account     **;

**  for new export level data.                                             **;

*****************************************************************************;

filename _all_ clear; libname _all_ clear;

options linesize=150 pagesize=58 nocenter nodate;

filename in 'd:\sas\state_exports\rawdata\';

filename in2 'd:\sas\state_exports\rawdata\SIC_2digit';

filename in3 'd:\sas\state_exports\rawdata\Employment';

libname out 'd:\sas\state_exports\rawdata';

%let yr1=97;

%let yr2=98;

%let goyr=96;

****************************************************************************************;

**                                                                                    **;

**                     TABLE 1 AND TABLE 2 AND TABLE 3                                **;

**                                                                                    **;

****************************************************************************************;

********;

**1997**;

********;

**PULL IN EXPORT LEVEL DATA FROM ITC**;

**  ASIAN 10 and MANUFACTURING ONLY**;

data asia10&yr1(keep=state country sic x_y&yr1) rest&yr1(keep=state country sic x_y&yr1);

  infile in2(st&yr1.om) firstobs=1;

  input @2 state $2. @7 sic 2. @12 country 4.  @18 x_y&yr1;

  **KEEP MANUFACTURING INDUSTRIES ONLY**;

    if 20<=sic<=39;

  **KEEP ASIA 10 COUNTRIES ONLY **;

    if country in (5700,5800,5820,5830,5880,5650,5570,5590,5490,5600) then output asia10&yr1;

       else output rest&yr1;

run;

**MAKE REAL ASIAN 10 EXPORTS with 1996 BASE**;

proc sort data=asia10&yr1; by sic; run;

proc sort data=out.xprice; by siccode; run;

data asia10&yr1(keep=state country sic r_y&yr1  x_y&yr1);

  merge asia10&yr1 out.xprice(rename=(siccode=sic) where=(sic<>0 and 20<=sic<=39));

  by sic;

  ch_p = ((y&yr1/y&goyr)-1)*100;

  r_y&yr1 = (x_y&yr1/((100+ch_p)/100));

run;

**MAKE ASIAN 10 EXPORTS BY STATE        **;

proc sort data=asia10&yr1; by state; run;

proc means data=asia10&yr1 sum noprint;

  var r_y&yr1 x_y&yr1;

  by state;

  output out=a10&yr1 sum=a10_&yr1 a10n_&yr1;

run;

**MAKE REAL WORLD LESS A-10 EXPORTS with 1996 BASE**;

proc sort data=rest&yr1; by sic; run;

proc sort data=out.xprice; by siccode; run;

data rest&yr1(keep=state country sic r_y&yr1  x_y&yr1);

  merge rest&yr1 out.xprice(rename=(siccode=sic) where=(sic<>0 and 20<=sic<=39));

  by sic;

  ch_p = ((y&yr1/y&goyr)-1)*100;

  r_y&yr1 = (x_y&yr1/((100+ch_p)/100));

run;

**MAKE REAL WORLD LESS A-10 EXPORTS BY STATE      **;

proc sort data=rest&yr1; by state; run;

proc means data=rest&yr1 sum noprint;

  var r_y&yr1 x_y&yr1;

  by state;

  output out=r&yr1 sum=r_&yr1 n_&yr1;

run;

********;

**1998**;

********;

**PULL IN EXPORT LEVEL DATA FROM ITC**;

**  ASIAN 10 and MANUFACTURING ONLY**;

data asia10&yr2(keep=state country sic x_y&yr2) rest&yr2(keep=state country sic x_y&yr2);

  infile in2(st&yr2.om) firstobs=1;

  input @2 state $2. @7 sic 2. @12 country 4.  @18 x_y&yr2;

  **KEEP MANUFACTURING INDUSTRIES ONLY**;

    if 20<=sic<=39;

  **KEEP ASIA 10 COUNTRIES ONLY **;

    if country in (5700,5800,5820,5830,5880,5650,5570,5590,5490,5600) then output asia10&yr2;

       else output rest&yr2;

run;

**MAKE REAL ASIAN 10 EXPORTS with 1996 BASE**;

proc sort data=asia10&yr2; by sic; run;

proc sort data=out.xprice; by siccode; run;

data asia10&yr2(keep=state country sic r_y&yr2  x_y&yr2);

  merge asia10&yr2 out.xprice(rename=(siccode=sic) where=(sic<>0 and 20<=sic<=39));

  by sic;

  ch_p = ((y&yr2/y&goyr)-1)*100;

  r_y&yr2 = (x_y&yr2/((100+ch_p)/100));

run;

**MAKE ASIAN 10 EXPORTS BY STATE        **;

proc sort data=asia10&yr2; by state; run;

proc means data=asia10&yr2 sum noprint;

  var r_y&yr2 x_y&yr2;

  by state;

  output out=a10&yr2 sum=a10_&yr2 a10n_&yr2;

run;

**MAKE REAL WORLD LESS A-10 EXPORTS with 1996 BASE**;

proc sort data=rest&yr2; by sic; run;

proc sort data=out.xprice; by siccode; run;

data rest&yr2(keep=state country sic r_y&yr2  x_y&yr2);

  merge rest&yr2 out.xprice(rename=(siccode=sic) where=(sic<>0 and 20<=sic<=39));

  by sic;

  ch_p = ((y&yr2/y&goyr)-1)*100;

  r_y&yr2 = (x_y&yr2/((100+ch_p)/100));

run;

**MAKE REAL WORLD LESS A-10 EXPORTS BY STATE      **;

proc sort data=rest&yr2; by state; run;

proc means data=rest&yr2 sum noprint;

  var r_y&yr2 x_y&yr2;

  by state;

  output out=r&yr2 sum=r_&yr2 n_&yr2;

run;

******************************************************************;

**PULL IN POPULATION DATA TO CREATE PERCAPITA NUMBERS IN TABLE 1**;

******************************************************************;

data pop;

  infile in3(pop97) firstobs=5;

  input state $2. pop97;

run;

**************************************;

**CREATE DATA for TABLES 1,2, and 3 **;

**************************************;

data x;

  merge a10&yr1 a10&yr2 r&yr1 r&yr2 pop;

  by state;

  if state in ("PR","VI") then delete;

  **Create world exports**;

    t_&yr1=a10_&yr1+r_&yr1;

    t_&yr2=a10_&yr2+r_&yr2;

  **Create change in world and a-10 exports**;

    t_ch = ((t_&yr2/t_&yr1)-1)*100;

    a10_ch = ((a10_&yr2/a10_&yr1)-1)*100;

  **Create A-10 share of world exports**;

    sh_&yr2 = (a10_&yr2/t_&yr2)*100;

    sh_&yr1 = (a10_&yr1/t_&yr1)*100;

  **Create per capita A-10 and world export levels**;

    pca10 = a10_&yr1/pop97;

    pctot = t_&yr1/pop97;

  **Create export levels in millions of $**;

    a10l&yr2 = a10_&yr2/1000000;

    tl&yr2 = t_&yr2/1000000;

    a10l&yr1 = a10_&yr1/1000000;

    tl&yr1 = t_&yr1/1000000;

  **Create state labels for tables**;

    staten=stnamel(state);

run;

****************************;

**CREATE RANKS FOR TABLE 1**;

****************************;

proc sort data=x(keep=state a10_&yr1) out=a10; by descending a10_&yr1; run;

proc sort data=x(keep=state t_&yr1) out=tot; by descending t_&yr1; run;

proc sort data=x(keep=state pca10) out=a10pc; by descending pca10; run;

proc sort data=x(keep=state pctot) out=totpc; by descending pctot; run;

data a10; set a10; ranka10 = _n_; run;

data tot; set tot; ranktot = _n_; run;

data a10pc; set a10pc; ranka10p = _n_; run;

data totpc; set totpc; ranktotp = _n_; run;

proc sort data=a10; by state; run;

proc sort data=tot; by state; run;

proc sort data=a10pc; by state; run;

proc sort data=totpc; by state; run;

************************************;

**COMBINE ALL TABLE DATA          **;

**  INCLUDING RANKS CREATED ABOVE **;

************************************;

data x;

  merge x a10 tot a10pc totpc;

  by state;

run;

****************************;

**PRINT TABLES 1, 2 and 3 **;

****************************;

proc sort data=x; by descending a10l&yr1; run;

proc print noobs;

  var staten a10l&yr1 ranka10 tl&yr1 ranktot pca10 ranka10p pctot ranktotp;

  format a10l&yr1 comma7. tl&yr1 comma7. pca10 comma5. pctot comma5.;

  title "Real(96$) Manufacturing Exports in 19&yr1";

  title2 "Asian-10 and Total";

  title3 "Table 1 -- 1997 Exports of Manufactured Good By State (millions of 1996 dollars)";

run;

proc sort data=x; by descending a10_ch; run;

proc print noobs;

  var staten a10_&yr1 a10_&yr2 a10_ch t_&yr1 t_&yr2 t_ch;

  format a10_ch 5.1 t_ch 5.1;

  title "Percentage Change in Total Real (96$) Manufacturing Exports from 19&yr1 to 19&yr2";

  title2 "Asian-10 and Total";

  title3 "Table 2 -- Change in State Exports of Manufactured Goods (percent change 1997-1998)";

run;

proc sort data=x; by descending sh_&yr1; run;

proc print noobs;

  var staten a10_&yr1 t_&yr1 sh_&yr1;

  format sh_&yr1 5.1;

  title "Real(96$) Manufacturing Exports to Asian 10 as Share of Total Manu. Exports in 19&yr1";

  title2 "Table 3 -- East Asia Share of State Exports of Manufactured Goods (1997)";

run;

****************************************************************************************;

**                                                                                    **;

**                          TABLE 4 and TABLE 5                                       **;

**                                                                                    **;

****************************************************************************************;

***************************************;

**PULL IN ASM - SHIPMENTS DATA CENSUS**;

**Currently this is 1996             **;

***************************************;

data manuship(keep=state sic0);

  infile in(asm_shipments) firstobs=10 obs=60 linesize=5000;

  input @1 state $2. sic0 sic20 sic21 sic22 sic23 sic24 sic25 sic26 sic27 sic28

                     sic29 sic30 sic31 sic32 sic33 sic34 sic35 sic36 sic37 sic38 sic39;

run;

proc sort data=manuship; by state; run;

proc transpose data=manuship out=manush;

  var sic0;

  by state;

run;

data manuship(drop=_name_);

  set manush;

  sic=substr(_name_,4,2)+0;

  rename col1=manuship;

run;

proc sort data=manuship; by state; run;

**************************************************;

**PULL IN EXPORT LEVEL DATA FROM ITC            **;

**  ASIAN 10 and MANUFACTURING ONLY             **;

**  FOR SAME YEAR AS SHIPMENTS DATA ABOVE       **;

**************************************************;

data x_y&goyr(keep=state country sic x_y&goyr);

  infile in2(st&goyr.om) firstobs=1;

  input @2 state $2. @7 sic 2. @12 country 4.  @18 x_y&goyr;

  **MANUFACTURING ONLY**;

    if 20<=sic<=39;

  **PICK ONLY ASIAN 10**;

    if country in (5700,5800,5820,5830,5880,5650,5570,5590,5490,5600);

run;

**MAKE ASIAN 10 EXPORTS BY STATE        **;

proc sort data=x_y&goyr; by state sic; run;

proc means data=x_y&goyr noprint;

  var x_y&goyr;

  by state;

  output out=x_a10&goyr sum=x_y&goyr.s;

run;

**********************************************************************;

**MAKE THE RATIO OF ASIAN 10 EXPORTS TO CURRENT DOLLAR SHIPMENTS    **;

**  MANUFACTURING EXPORTS / MANUFACTURING SHIPMENTS                 **;

**********************************************************************;

data mship&goyr(drop=_freq_ _type_);

  merge x_a10&goyr manuship;

  by state;

  if state in("PR","VI") then delete;

  **Make ratio var**;

    ratio = ((x_y&goyr.s/1000)/manuship)*100;

  **Make state labels for table**;

    staten = stnamel(state);

run;

*********************;

**PRINT TABLE 4    **;

*********************;

proc sort data=mship&goyr; by descending ratio; run;

proc print data=mship&goyr noobs;

  var staten x_y&goyr.s manuship ratio;

  format ratio 5.1;

  title "Ratio of Asian-10 Total Manufacturing Exports to Total Manufacturing Shipments";

  title2 "19&goyr";

  title3 "Table 4 - Manufactured Exports to East Asia as a Share of Manufactured Shipments (by state)";

run;

**********************************************************************;

**CREATE EFFECT ON OUTPUT FOR TABLE 5                               **;

**  USING REAL EXPORT CHANGE * RATIO OF EXPORTS TO SHIPMENTS        **;

**********************************************************************;

proc sort data=x; by state; run;

proc sort data=mship&goyr; by state; run;

data effect;

  merge x(keep=state a10_ch) mship&goyr(keep=state ratio);

  by state;

  **Make effect var**;

    eff = ((a10_ch/100) * (ratio/100))*100;

  **Make state labels for table**;

    staten = stnamel(state);

run;

*********************;

**PRINT TABLE 5    **;

*********************;

proc sort data=effect; by descending eff; run;

proc print data=effect noobs;

  var staten ratio a10_ch eff;

  format eff 6.2;

  title "Table 5 - The Effect of the Trade Shock on Manufacturing Output (shipments)";

run;

proc means data=effect;

  var eff;

  title "Unweighted Average of Table 5";

run;

****************************************************************************************;

**                                                                                    **;

**                                    TABLE 6                                         **;

**                                                                                    **;

****************************************************************************************;

*****************;

**1997 and 1998**;

*****************;

******************************************;

**MAKE REAL ASIAN 10 by COUNTRY by STATE**;

******************************************;

proc sort data=asia10&yr1; by country state; run;

proc means data=asia10&yr1 sum noprint;

  var r_y&yr1;

  by country state;

  output out=indiv&yr1 sum=indiv_&yr1;

run;

proc sort data=asia10&yr2; by country state; run;

proc means data=asia10&yr2 sum noprint;

  var r_y&yr2;

  by country state;

  output out=indiv&yr2 sum=indiv_&yr2;

run;

***********************************************************************;

**MAKE ASIAN 10 EXPORTS CHANGE FROM 1997 TO 1998 BY COUNTRY BY STATE **;

***********************************************************************;

proc sort data=indiv&yr2; by state; run;

proc sort data=indiv&yr1; by state; run;

data x(keep=staten country change neg);

  merge indiv&yr2 indiv&yr1;

  by state;

  **Delete non-states**;

    if state in ("PR","VI") then delete;

  **Change in levels**;

    change = ((indiv_&yr2/indiv_&yr1)-1)*100;

  **Tag declining exports**;

    if change<0 then neg = 1; else neg = 0;

  **Make state labels for table**;

    staten=stnamel(state);

run;

/*

**View Changes by Country by State -- in Table format**;

proc sort data=x; by staten; run;

proc transpose data=x out=xt1;

  var change;

  by staten;

  id country;

  idlabel country;

run;

*/

***********************************************************************;

**COUNT NUMBER OF NEGATIVE CHANGES IN ASIAN 10 FOR EACH STATE        **;

***********************************************************************;

proc sort data=x; by staten; run;

proc means data=x sum noprint;

  var neg;

  by staten;

  output out=t6 sum=;

run;

*******************************;

**PRINT TABLE 6 DATA         **;

*******************************;

proc print data=t6 noobs;

  var staten neg;

  title "Table - 6 Declining Exports to East Asia on a Country Basis";

run;

Run the Regression Reported in the Text

Program File:


review_regress.sas

filename _all_ clear; libname _all_ clear;

filename in 'd:\sas\state_exports\rawdata';

libname in2 'd:\sas\state_exports\rawdata';

***********************************************************************************************;

**                                     EFFECT OF OIL SHOCK                                   **;

***********************************************************************************************;

data oilshock;

  infile in(oilshock) firstobs=6;

  input @1 state $2. @28 oil1992 @40 oil2000;

run;

***********************************************************************************************;

**                                        EMPLOYMENT GROWTH                                  **;

***********************************************************************************************;

filename _all_ clear; libname _all_ clear;

filename in 'd:\sas\state_exports\rawdata\Employment';

libname out 'd:\sas\state_exports\rawdata';

data manuemp;

  length state $2.;

  infile in(manuemp_bystate_nsa) firstobs=545 obs=597 linesize=5000 dlm=",";

  input varname $ e1-e10;

fips=substr(varname,4,2);

state=fipstate(fips);

if state="--" then delete;

  keep state fips e1-e10;

**Non-continuous series**;

if fips=56 then e8=10.8;

run;

data manuemp(keep=state megt megt_1 megt_2 megt_3 megtrend);

  set manuemp;

  rename e1=ann89 e2=ann90 e3=ann91 e4=ann92 e5=ann93

         e6=ann94 e7=ann95 e8=ann96 e9=ann97

         e10=ann98;

**CREATE PERCENT CHANGES**;

MEGt_3=((e7/e6)-1)*100;

MEGt_2=((e8/e7)-1)*100;

MEGt_1=((e9/e8)-1)*100;

MEGt=((e10/e9)-1)*100;

MEGTREND = mean(MEGt_1,MEGt_2,MEGt_3);

run;

***********************************************************************************************;

**                                 ASIA EFFECT ON SHIPMENTS                                  **;

***********************************************************************************************;

filename _all_ clear; libname _all_ clear;

filename in 'd:\sas\state_exports\rawdata\';

filename in2 'd:\sas\state_exports\rawdata\SIC_2digit';

libname out 'd:\sas\state_exports\rawdata';

%let goyr=96;

%let yr1=97;

%let yr2=98;

**************************************;

**1996                              **;

**PULL IN ASM - SHIPMENTS DATA CENSUS*;

**************************************;

data manuship(keep=state sic0);

  infile in(asm_shipments) firstobs=10 obs=60 linesize=5000;

  input @1 state $2. sic0 sic20 sic21 sic22 sic23 sic24 sic25 sic26 sic27 sic28

                     sic29 sic30 sic31 sic32 sic33 sic34 sic35 sic36 sic37 sic38 sic39;

run;

proc sort data=manuship; by state; run;

proc transpose data=manuship out=manush;

  var sic0;

  by state;

run;

data manuship(drop=_name_);

  set manush;

  sic=substr(_name_,4,2)+0;

  rename col1=manuship;

run;

proc sort data=manuship; by state; run;

**************************************************;

**PULL IN EXPORT LEVEL DATA FROM ITC            **;

**  ASIAN 10 and MANUFACTURING ONLY             **;

**  FOR SAME YEAR AS SHIPMENTS DATA ABOVE       **;

**************************************************;

data x_y&goyr(keep=state country sic x_y&goyr);

  infile in2(st&goyr.om) firstobs=1;

  input @2 state $2. @7 sic 2. @12 country 4.  @18 x_y&goyr;

  **GET RID OF CATEGORIES WITH NO MATCHING OUTPUT VALUE**;

    if sic > 39 or sic < 20 then delete;

  **PICK ONLY ASIAN 10**;

    if country in (5700,5800,5820,5830,5880,5650,5570,5590,5490,5600);

run;

**MAKE ASIAN 10 EXPORTS BY STATE        **;

proc sort data=x_y&goyr; by state sic; run;

proc means data=x_y&goyr noprint;

  var x_y&goyr;

  by state;

  output out=x_a10&goyr sum=x_y&goyr.s;

run;

**********************************************************************;

**MAKE THE RATIO OF ASIAN 10 EXPORTS TO CURRENTT DOLLAR SHIPMENTS   **;

**  MANUFACTURING EXPORTS / MANUFACTURING SHIPMENTS                 **;

**********************************************************************;

data mship&goyr(drop=_freq_ _type_);

  merge x_a10&goyr manuship;

  by state;

  ratio = ((x_y&goyr.s/1000)/manuship)*100;

  if state in("PR","VI") then delete;

run;

**1997**;

**************************************************;

**PULL IN EXPORT LEVEL DATA FROM ITC            **;

**  ASIAN 10 and MANUFACTURING ONLY             **;

**THIS MAKES THE LATEST CHANGE DATA (97 TO 98)  **;

**************************************************;

data x_y&yr1(keep=state country sic x_y&yr1);

  infile in2(st&yr1.om) firstobs=1;

  input @2 state $2. @7 sic 2. @12 country 4.  @18 x_y&yr1;

  **GET RID OF CATEGORIES WITH NO MATCHING OUTPUT VALUE**;

    if sic > 39 or sic < 20 then delete;

  **PICK ONLY ASIAN 10**;

    if country in (5700,5800,5820,5830,5880,5650,5570,5590,5490,5600);

run;

*********************************************;

**MAKE REAL ASIAN 10 EXPORTS with 1996 BASE**;

*********************************************;

proc sort data=x_y&yr1; by sic; run;

proc sort data=out.xprice; by siccode; run;

data x_y&yr1;

  merge x_y&yr1 out.xprice(rename=(siccode=sic) where=(sic<>0 and 20<=sic<=39));

  by sic;

  ch_p = ((y&yr1/y&goyr)-1)*100;

  r_y&yr1 = (x_y&yr1/((100+ch_p)/100));

run;

******************************************;

**MAKE ASIAN 10 EXPORTS BY STATE        **;

******************************************;

proc sort data=x_y&yr1; by state; run;

proc means data=x_y&yr1 sum noprint;

  var r_y&yr1;

  by state;

  output out=x_a10&yr1 sum=x_y&yr1.s;

run;

**1998**;

**************************************************;

**PULL IN EXPORT LEVEL DATA FROM ITC            **;

**  ASIAN 10 and MANUFACTURING ONLY             **;

**************************************************;

data x_y&yr2(keep=state country sic x_y&yr2);

  infile in2(st&yr2.om) firstobs=1;

  input @2 state $2. @7 sic 2. @12 country 4.  @18 x_y&yr2;

  **GET RID OF CATEGORIES WITH NO MATCHING OUTPUT VALUE**;

    if sic > 39 or sic < 20 then delete;

  **PICK ONLY ASIAN 10**;

    if country in (5700,5800,5820,5830,5880,5650,5570,5590,5490,5600);

run;

*********************************************;

**MAKE REAL ASIAN 10 EXPORTS with 1996 BASE**;

*********************************************;

proc sort data=x_y&yr2; by sic; run;

proc sort data=out.xprice; by siccode; run;

data x_y&yr2;

  merge x_y&yr2 out.xprice(rename=(siccode=sic) where=(sic<>0 and 20<=sic<=39));

  by sic;

  ch_p = ((y&yr2/y&goyr)-1)*100;

  r_y&yr2 = (x_y&yr2/((100+ch_p)/100));

run;

******************************************;

**MAKE ASIAN 10 EXPORTS BY STATE        **;

******************************************;

proc sort data=x_y&yr2; by state; run;

proc means data=x_y&yr2 sum noprint;

  var r_y&yr2;

  by state;

  output out=x_a10&yr2 sum=x_y&yr2.s;

run;

******************************************;

**MAKE ASIAN 10 EXPORTS CHANGE BY STATE **;

******************************************;

data x_a10;

  merge x_a10&yr1 x_a10&yr2;

  by state;

  change = ((x_y&yr2.s/x_y&yr1.s)-1)*100;

  if state in ("PR","VI") then delete;

run;

**********************************************************************;

**MAKE THE EFFECT on THE STATE MANUFACTURING SECTOR                 **;

**  MANU EXPORTS CHANGE  * MANUFACTURING RATIO                      **;

**********************************************************************;

proc sort data=mship&goyr; by state; run;

proc sort data=x_a10; by state; run;

data effect(keep=state ASIAMAN);

  merge x_a10 mship&goyr;

  by state;

  ASIAMAN = ((change/100)*(ratio/100))*100;

  if state in("PR","VI") then delete;

run;

*************************;

**COMBINE ALL DATA     **;

*************************;

proc sort data=oilshock; by state; run;

proc sort data=manuemp; by state; run;

proc sort data=effect; by state; run;

data allreg;

  merge oilshock manuemp effect;

  by state;

run;

***********************************************************************************************;

**                                     RUN REGRESSION                                        **;

***********************************************************************************************;

proc reg data=allreg;

  model MEGT = MEGTREND ASIAMAN OIL2000;

  title "Manu Employment = Various Lags of Manu Employment, Asia-10 Effect of Manu GSP,";

  title2 "      Oil Shock on 2000 NonAg Employment";

run;

quit;

This is the program file for Tables 1-6 and the regression analysis in “State Exports and the Asian Crisis” by Patricia S. Pollard and Cletus C. Coughlin in the January/February 2000 issue of the Federal Reserve Bank of St. Louis Review.


This file contains:


SAS v6.12 programs to create tables and regression results











